Using the "HP-71B" hand-held computer for data entry while running first order, class II level surveys

BY

Kenneth M. Yamashita

Open-file Report 89-179

This report is preliminary and has not been reviewed for conformity with U. S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS.

¹David A. Johnston Cascades Volcano Observatory
5400 MacArthur Blvd., Vancouver, Washington 98661
## CONTENTS

<table>
<thead>
<tr>
<th>Section</th>
<th>Page</th>
</tr>
</thead>
<tbody>
<tr>
<td>INTRODUCTION</td>
<td>3</td>
</tr>
<tr>
<td>USING THE HP-71B</td>
<td>4</td>
</tr>
<tr>
<td>LOADING PROGRAMS FROM THE HP-71B TO THE DIGITAL CASSETTE RECORDER</td>
<td>4</td>
</tr>
<tr>
<td>LOADING PROGRAMS INTO THE HP-71B FROM THE CASSETTE RECORDER OR CARD READER</td>
<td>4</td>
</tr>
<tr>
<td>LOADING PROGRAMS FROM THE HP-71B TO A MAGNETIC CARD</td>
<td>5</td>
</tr>
<tr>
<td>INSTRUCTIONS FOR RUNNING PROGRAM &quot;STADLEV&quot;</td>
<td>5</td>
</tr>
<tr>
<td>PRINTING THE DATA FROM INTERNAL RAM</td>
<td>9</td>
</tr>
<tr>
<td>WRITING THE DATA TO THE DIGITAL CASSETTE RECORDER</td>
<td>9</td>
</tr>
<tr>
<td>LOADING DATA FROM THE HP-71B TO A MAIN FRAME COMPUTER VIA THE RS232C INTERFACE</td>
<td>10</td>
</tr>
<tr>
<td>ENTERING A PROGRAM INTO THE HP-71B MANUALLY</td>
<td>12</td>
</tr>
<tr>
<td>PROGRAM LISTINGS</td>
<td>13</td>
</tr>
<tr>
<td>PROGRAM STADLEV</td>
<td>13</td>
</tr>
<tr>
<td>PROGRAM RECALL</td>
<td>16</td>
</tr>
<tr>
<td>PROGRAM TAPEWRT</td>
<td>16</td>
</tr>
<tr>
<td>PROGRAM TAPEDUMP</td>
<td>17</td>
</tr>
<tr>
<td>PROGRAM TAPERD</td>
<td>18</td>
</tr>
<tr>
<td>LIST OF VARIABLES FOR STADLEV PROGRAM</td>
<td>19</td>
</tr>
<tr>
<td>CONCLUSION</td>
<td>21</td>
</tr>
<tr>
<td>ACKNOWLEDGEMENTS</td>
<td>21</td>
</tr>
<tr>
<td>REFERENCES</td>
<td>21</td>
</tr>
</tbody>
</table>
INTRODUCTION:

Initially this document was put together as a field guide for the author's survey crew while running first order level surveys, but because of the ease of operation of the system and the time saved by this system, it was decided to open-file this report. The data logging program is "STADLEV". The program is designed to be used as an electronic field notebook to record data while running first order, class II level surveys. The program is written in a language very similar to BASIC, but with subtle differences unique to the HP-71B computer. The program can be easily modified to accept data for class 1 or second order level surveys by changing certain parameters in the program (see NOAA publication S/T 81-29).

Additional programs required to complete the data-gathering process are:

1. RECALL. This program recalls the data from internal RAM of the HP-71B and prints a hard copy.

2. TAPEWRT. This program transfers the data from internal RAM to a Digital Cassette Recorder.

3. TAPEDUMP. This program retrieves the data stored on the Digital Cassette recorder and transfers it to a main frame computer using an RS232C interface.

4. TAPERD. Although not essential to the data gathering process, Taperd allows you to see what is stored on the cassette tape.

Although it is not critical to the operation of the system, it is recommended that a minimum of one 4K - memory module be installed and that a printer be used to preserve the data. Using a digital cassette recorder or disc drive would save time if the data were to be reduced by a main frame computer. If any peripherals (eg. Digital Cassette Recorder) are to be used with the HP-71B, an HP-IL interface is needed. The programs described in this report presume that the HP-71B system has the following:

1. HP-71B hand-held computer
2. HP 82162A Thermal Printer
3. HP 82161A Digital Cassette Drive
4. HP 82420A 4K Memory Module
5. HP 82401A HP-71 HP-IL Interface Module
6. HP 82164A HP-IL/RS-232C Interface
7. HP 82400A Card Reader
8. HP 82707A Blank Card Pac
9. HP 82175A Thermal Paper
10. HP 82176A Mini Cassette Tape
11. Black Box (ABC Monitor by Micom co.)
Everyone who uses the HP-71B should read the owners manual and be familiar with the 71B's operation.

**USING THE HP-71B:**

1. Turn the HP-71B on by pressing the [ON] key.
2. Turn the HP-71B off by pressing the yellow [f] key then the [ON] key.
3. Clear the display by pressing the [ON] or (ATTN) key.
4. Delete a character to the right of the cursor by pressing the yellow [f] key then the [>] (-char) key.
5. Delete a character to the left of the cursor by pressing the yellow [f] key then the [<] (back) key.
6. Since the HP-71B can only display 21 characters at a time on the display screen, press the blue [g] key, then either the [<] (back) or [>] (-char) (depending on which direction you need to look at) to see the entire display.

**LOADING PROGRAMS FROM THE HP-71B TO THE DIGITAL CASSETTE RECORDER**

Be sure that the HP-71B is off before connecting a peripheral. Connect the Digital Cassette Recorder to the HP-71B. Turn on the Cassette Recorder first, before the HP-71B, then the HP-71B. The command to write to the digital cassette recorder is COPY PROGRAMNAME TO :TAPE <END LINE>.

**LOADING PROGRAMS INTO THE HP-71B FROM THE CASSETTE RECORDER OR CARD READER**

If the programs are to be loaded from a digital cassette recorder the command is COPY PROGRAMNAME :TAPE <END LINE>.

If the programs are to be read from the card reader, the command is COPY CARD <END LINE>. The 71B will prompt [Read: Align then ENDLN]. Insert the card into the card reader slot then enter the <END LINE> key, the 71B will prompt [Pull card]; pull the card through the card reader. If there is more then one track to be copied the 71B will display [TRK 1 done] and will prompt [Read: Align then ENDLN]. Insert track 2 of the card, enter the <END LINE> key and pull the card through the card reader, the 71B will prompt [pull 2 of 3], the 71B will display [TRK 2 done], [Read Align then ENDLN], insert track 3, the 71B will prompt [Pull 3 of 3], enter the <END LINE> key and pull the card through the card reader, the 71B will display [TRK 3 Done] and the cursor [>] will appear. When this message appears, this indicates the program has been successfully loaded.
LOADING PROGRAMS FROM THE HP-71B TO A MAGNETIC CARD

The command to write to a program card is COPY PROGRAMNAME TO CARD <END LINE>. Insert the card into the card reader slot. At the prompt [Wrt: then ENDLN] enter <END LINE>; the prompt [Pull 1 of 2] will be displayed. Pull the card through the card reader. The 71B will prompt [VFY :Align ENDLN]. Insert the same track of the card to verify that the program was copied correctly. Enter <END LINE> and pull the card through the reader. The 71B will display [TRK 1 Done] [Wrt:Align then ENDLN]. Repeat the process as before. When the program is successfully loaded, the message [TRK 2 Done], or [TRK 3], (depending on how many tracks there are), will be displayed and the cursor [>] will appear.

NOTE: The HP-71B shuts off automatically in 10 minutes if no keys are pressed in that time; this function has been suppressed in STADLEV program and will not be a problem. In the event that the HP-71B does turn off, you can continue running the program by first turning the HP on, then pressing the yellow [ ] key, then the [ +] (cont) key. DO NOT PRESS THE [RUN] KEY or type RUN to continue the program. Doing so will cause the program to start from the very beginning and not where the program stopped.

INSTRUCTIONS FOR RUNNING PROGRAM "STADLEV"

All characters highlighted in BOLD are to be entered by the operator and all characters in brackets [ ] are messages from the HP-71B program.

All input is entered by using the <END LINE> key. You will need a printer and a cassette recorder to store the level data. If there is a need to format a new tape, see the instructions under the heading "WRITING DATA TO THE CASSETTE RECORDER".

1. To activate the program type RUN STADLEV <ENDLINE>.

2. [Do you really want to run Program Stadlev? Y, N]. When program "STADLEV" is called, all files and variable names in internal ram are purged to ensure that data from previous surveys do not contaminate the current data. If a mistake is made and program STADLEV is accidently run before the data in internal RAM are printed or written to the cassette recorder, data from the previous survey will be destroyed. This means that the previous run will be lost and will need to be repeated. This prompt allows the operator to enter an N, which ends the program without destroying the previous data set. If a Y is entered the program will prompt:

3. [Enter gun ID]. Enter the serial number of the level gun. (It is located under the front lens of the NA2). <END LINE>. Hereafter <END LINE> will not be indicated.

4. [Type of level gun]. Enter the manufacturer and model. i.e. Wild NA2.

5. [Temp in C or F]. Enter either the letter C or F, depending on whether the
temperatures are taken in degrees celsius or fahrenheit.

6. [Rod type Wild or Kern]. The offset for Kern rods is 301.500 and for Wild rods is 301.550. The program will know which value to use from the input. The spelling of the rod type must be correct or the program will not accept the input.

7. [Enter Network]. This should be a simple name such as Yellowstone, or Newberry, etc. A maximum of 15 characters.

8. [Percent Cloud Cover]. A ball park figure is sufficient. If 35% then enter 35.

9. [Wind Direction & speed]. Again a ball park figure is sufficient, such as ENE35 or NW10.

10. [Height of gun person in M.] This will be used to determine the true difference between the level gun and the rod readings so enter the number to the nearest tenth. e.g. 1.9.

11. [Enter both Rod #'S]. Enter both the A and B rod identification numbers such as 5140A 5140B. Separate the rod ID's by a space.

12. [Stadia Imbalance is XX ]. The stadia imbalance, if any, is displayed for verification. This will be "0" at the start of a run.

13. [Is XX correct? Y,N]. If the value displayed is not correct it can be corrected now; If correct enter a Y, otherwise enter a N.

14. [BS Bench mark]. You are allowed up to 10 characters for the bench mark identification. Enter it as it is stamped on the bench mark.

15. [BS ROD ID]. Enter the rod identification that is on the backsight benchmark.

16. [Backsight low]. This will be the backsight low scale middle wire, read to 3 decimal places. Enter all readings with a decimal point.

16a. [Bs top stadia]. Enter the top backsight stadia wire reading to one decimal place.

16b. [Bs bottom stadia]. Enter the backsight bottom stadia wire reading, also to one decimal place.

If the stadia difference between the top and middle, and middle and bottom wires is greater then 0.5 meter, (this is arbitrary and is meant to catch observational or input blunders) the numbers will be rejected, beep and cycle back to step 16a [Bs top stadia]. If the difference is still bad after the second try, the computer will beep twice and cycle back to step 16 [Backsight low]. If the check is good, the distance in meters between the instrument and the backsight rod will be displayed, followed by the following prompts:
17. [Foresight low]. This will be the foresight low scale middle wire, read to 3 decimal places.

17a. [Fs top stadia]. Enter the foresight top wire reading to one decimal place.

17b. [Fs bottom stadia]. Enter the bottom stadia wire reading to one decimal place.

Again, the difference between the three wires must be less then 0.5 meter or the numbers will be rejected and the program will cycle back to step 17a [Fs top stadia]. If the check is good, the distance between the instrument and the foresight rod will be displayed. Again as with the backsight, if the stadia difference does not check after the second try the computer will beep twice and cycle back to step 17 [Foresight low].

The total imbalance between the cumulative backsight stadia and the cumulative foresight stadia cannot exceed 10 meters. The individual stadia imbalance between the backsight and foresight cannot exceed 5 meters (see NOAA publication S/T 81-29). If either of these exceeds the 5 or 10 meter envelope the computer will beep, reject the entire set-up, display the message [XX Stadia] XX being the imbalance, and cycle back to step 16 [Backsight low]. If all parameters check, the following prompts will be displayed.

18. [Foresight high]. Enter the Foresight high scale, middle wire reading.

19. [Backsight high]. Enter the Backsight high scale, middle wire reading.

After all readings for the set-up are made, the elevation difference is calculated between the low scale readings and the high scale readings. A maximum height difference of 0.030 cm (see NOAA publication S/T 80-98) is allowed between the low and high scale elevation. If the difference is greater then this, a beep will sound and the message [WARNING BAD READING] will be displayed. If this happens, the low scale and high scale readings will be rejected and will have to be repeated; however, if the stadia balance was good the program will not ask for the stadia readings again.

20. [Enter Temp Top, Bottom]. Both temperatures will be entered as one value, top temperature first, then bottom temperature, separated by a decimal point. For example, if the temperatures are 27.0 and 28.3, respectively, they should be entered as 270.283. If the temperatures are 10.0 and 9.0 then enter the data as 100.090. Always have 3 digits after the decimal point. No decimal point follows the last digit.

If all the checks are good, a number (should be <= 10) and the message [Foresight is XX long] or [Foresight is XX short] (depending on which it is) will be displayed. This number (the stadia imbalance) is the sum of the differences between the total backsight stadia and total foresight stadia, and the imbalance should be corrected accordingly. Individual set-up imbalance must be less then 5 meters (the program will reject anything larger then this).

The program will now display the message [Stadia distance is XX.X], XX.X being
the cumulative stadia distance. The message [XXX Bytes left] will also be displayed. This indicates how much memory is left in the computer. The heading for the section takes about 200 bytes, and each set-up takes about 130 bytes. This means that 20 set-ups, including header, would require about 2800 bytes (173 records, explained latter).

21. [Was Fs on a BM? Y, N]. If the foresight rod is on a bench mark the answer would be a "Y" (yes). If a Y is entered the program will prompt:

21a. [Enter Fsbm ID]. Enter the bench mark identification as stamped on the monument.

The program will now display [Elev diff is XXXX], [XXX Bytes left]. Also displayed will be the stadia distance, and the stadia imbalance. If the number of bytes displayed is less then 2000 the data should be written to the cassette recorder before continuing. If not, you will probably run out of storage space before you reach the next bench mark.

21b. [Continue or End? C, E]. If an E is entered the program will end, and the message [Program ended] will be displayed. IF YOU END THE PROGRAM YOU MUST DUMP THE DATA. Remember that the data will be purged when program "STADLEV" is called again.

If you answer C (continue) to question number 21b [Continue or end? C, E], the program will prompt:

21c. [Was FS a BM or TBM]. Enter either BM or TBM. If a TBM (temporary bench mark), the program will cycle to step 12. The difference is that when on a BM the stadia distance and elevations are "zeroed" out, but on a TBM the stadia distance and elevations does not "zero" out but keeps accumlating.

The program will now cycle back to step number 12 and start the whole sequence again.

12. [Stadia imbalance is XX] This number should be the same as the number displayed between steps 20 and 21.

If the answer to 21c was N, the program will cycle to step 3.

3. [Enter Gun ID]. Continue as before.

If you answer N (no) to number 21 [Was Fs on a BM? Y, N], the program will cycle back to step number 16 [Backsight low].

NOTE: A wrong value accidently entered while running the program can be cleared out by entering numbers that will not satisfy the checks. The best way to cycle through the program is to enter all zero's for the input.
PRINTING THE DATA FROM INTERNAL RAM

BE SURE THAT THE HP-71B IS "OFF" BEFORE CONNECTING THE PRINTER.

Connect the printer to the HP-71B. Then turn on the printer before the HP-71B. Type **RUN RECALL** <END LINE>. The printer should be printing the stored values. The second to the last number (sandwiched between the the "End of file" error message and the time of day) printed on the paper tape is the number of records. Record this number because it will be needed when the data are dumped to the cassette recorder (explained in the section under Run Tapewrt).

WRITING THE DATA TO THE DIGITAL CASSETTE RECORDER

Again as with the printer, be sure that the cassette recorder, or any other peripheral device, is turned on first, before the HP-71B.

Before data can be written to the cassette tape, the tape must first be formatted. Because of the time involved in formatting the tape, and excessive battery consumption, this should be done before going into the field.

When the cassette recorder is first turned on it will cycle forward and backward; this is normal. Do not do anything until the red "Busy" light goes off. The command to format the tape is **INITIALIZE:TAPE,1** <END LINE>; this configures the tape to contain only 1 file. It will take about 5 minutes to complete the initialization process. When the red "Busy" light goes out type **CREATE DATA "LEV1:TAPE",3000,16** <END LINE>. This says CREATE a DATA file (as opposed to a text file), call this file on TAPE LEV1, and allow 3000 records (this is the maximum for a tape), each of which will be 16 bytes long. This takes about 10 minutes, so be sure that the cassette recorder is fully charged or plugged into the charger unit.

To write the data to the cassette recorder type **RUN TAPEWRT**. The cassette will cycle forward and backward; and the prompt [First write to tape? Y,N] will be displayed. The question is, is this the first time that you are writing to this tape? It is imperative that you understand what you need to do next. Since the tape has been set to be one continuous file, we need to keep track of where the next data point will start writing on the tape. The first time that data are written to the tape they will start writing at record 000. The first write will be at record 000. If you enter a Y (yes), the message [Do you really mean yes? Y,N] will be displayed. This allows you to change your mind if it is incorrect. If you answer Y again, the cassette recorder will automatically reset the file pointer to record 000 (this is why you don't want to say Y when you mean N) and start to record the data from the HP-71B to the cassette recorder. If you enter N (no), the program will get a value from internal ram, and the message [The last record was XXX] [Is this the correct record number to write? Y,N] will be displayed. If you enter N (no), the HP will prompt [ENTER THE CORRECT RECORD].

How do you know what the correct record number should be? Recall that when the data were printed out, the second to the last number on the paper tape, after the "End of
file" message and before the time, is the number of records printed. (see the section about Run Recall). This is where we use that number. If the number on the paper tape was 105, that means that the first time data were written to the cassette tape they started from records 000 to 104. If the "pointer" is now at record number 105, the second time data are written to the cassette they will start writing record number 105-1 (104). If the second write had 120 records, the new record number would be 104+120=224. The third write would start at record 224. Giving it a lower number than 224 will result in the data being written over. We did write over record 104, or the last data point, because we don't really need the number of records printed on the paper tape, to be stored on the cassette tape, and our experience has shown that "blank" spaces on the cassette tape can be interpreted as an "End of File" marker, making it difficult to retrieve the data. The time of day printed on the paper tape is not written to the file.

When all of the data have been written to the cassette recorder, a number will be displayed. This number is the last legitimate record on the tape. This is also the number that should be displayed when "TAPEWRT" program is run again.

A notebook with the benchmark names and number of records should be used to keep track of the number of records for each section run. The sum of all the record numbers will be needed when running program "TAPEDUMP".

A note of caution: The heading for each section run takes up 13 records and each set-up thereafter requires 8 records. This means that when the record number displayed is 2800, there is enough room for approximately 20 set-ups. Don't attempt to write on to the same tape if the number of records approaches this number; the program is made so it won't accept an input number larger than 2700, although it is capable of writing up to record 3000.

Another tape should be used to dump the data to the cassette recorder once the number of records approaches 2700. Do not confuse bytes with record number. Each record contains 16 bytes, and each data point is one record.

LOADING DATA FROM THE HP-71B TO A MAIN FRAME COMPUTER VIA THE RS232C INTERFACE

The instructions for "Loading" are given for interfacing with a VAX 11-750 main frame computer. Other computers may require different commands.

The steps required before "Loading" are:

2. "Log" into the Main frame.
3. Connect the RS232 to a "Black box".
4. Using the function keys or by typing (depends on your terminal):
   4a. SET TERM/SPEED = 4800
   4b. SET SPEED (Baud rate) on main frame to 4800
   4c. PARITY = NONE

10
4d. STOP BIT = 1  
4e. CHAR BIT = 8  
4f. AUTO XOFF AT BUFFER TO 64  
5. From the main frame terminal type CREATE Filename.  
6. On the HP-71B, type RUN TAPEDUMP.  

The HP-71B will prompt [Enter first record to wrt]. In most cases the first record will be 000, but you can start with any record number. The next prompt will be [Enter last record to wrt]. Since the tapes will most likely be used for different surveys, data from previous assignments could still be on the tapes. If the previous survey had more data points than the current survey, data from the prior survey would also be written to the main frame.  

This is because the program will not know where the last "legitimate" data point is and where the old data begin. Indicating the last record that should be written will prevent unwanted data from being written to the main frame file. This LAST RECORD will be the sum of all of the individual record numbers written on each tape (maximum of 3000 per tape, realistically 2700) derived when running program "TAPEWRT".  

When all the data have been dumped:  

From the main frame terminal close the created file.  

The data written to the file will contain one data point per line. eg:  

88/08/15 !Year, Month, and Day  
09:45:25 !Hour, Minute, and Seconds  
123456 !Gun serial number  
WILD NA2 !Gun manufacture and type of level  
C !Temperature units  
WILD !Rod manufacture identification  
NEWBERRY !Network  
25% !Percentage of cloud cover  
ENE10 !Wind direction and speed  
1.85 !Height of instrument person  
5140A 5140B !Serial numbers of both rods  
CVO85-200 !Backsight bench mark  
5140A !Rod ID on backsight bench mark  
100.000 !Backsight low scale middle wire  
401.550 !Backsight high scale middle wire  
27.5 !Backsight stadia distance  
75.00 !Foresight low scale middle wire  
376.550 !Foresight high scale middle wire  

*Note that the backsight low scale and high scale were the first and last to be read but follow each other when printed.  

29.0 !Foresight stadia distance
22.21 !Top and bottom temperatures
-25.00 !Cumulative elevation difference in cm between bench marks

*If the next set-up is not a benchmark, this space would have the backsight low scale reading. If it is a benchmark, the next sequence will be printed.

CVO85-202 !Foresight bench mark
56.5 !Total stadia distance from previous bench mark
1.5 !Total stadia imbalance at the bench mark. If a positive value the foresight is long. If a negative value the foresight is short

*The next three items are printed on the paper tape but are not written to the cassette tape when program TAPEWRT is run.

End of File !Error message from the HP-71B indicating no more data
23 !This is the number of records printed. This is the number needed for the "Record number" when running program TAPEWRT.
10:32:10 !The time of day that this tape was printed

*If the survey was continued, that is if the data were not written to tape, this space would start with the Year, Month, and day.

ENTERING A PROGRAM INTO THE HP-71B MANUALLY:

Before entering the program you need to enter:

1. Edit programname <End Line>
2. Auto !This will automatically number the lines for you.

All instructions following the ! mark (except for line "10") are comment statements to explain the logic of the program, and are not meant to be written as part of the program.
PROGRAM LISTINGS

PROGRAM STADLEV

10 ! PROGRAM STADLEV
20 DELAY 1,0.1 !Delay 0.1 seconds between characters display
30 DISP 'DO YOU REALLY WANT TO RUN PROGRAM STADLEV? Y,N';
40 INPUT FS$
50 IF FS$# 'Y' THEN 940 !If not Y then end the program, else continue
60 ASSIGN #1 TO LEV @ SFLAG -3 !Disables the automatic time out
70 DESTROY ALL !Clears out all variables
80 PURGE LEV !Cleans the file out
90 ASSIGN #1 TO LEV !Open the channel to file "lev"
100 DIM E(50),E$(50) @D = 0 @D1 = 0 @D2 = 0
110 G = .030 @P = .5 @E(1) = 0 @X = 0
120 INPUT 'ENTER GUN ID ';G1$ !Enter the level gun serial number
130 INPUT 'TYPE OF LEVEL GUN';G9$ !Gun manufacture and gun model
140 INPUT 'TEMP IN C OR F? ';T$ !Enter the temperature units
150 INPUT 'ROD TYPE WILD OR KERN ';R$ !Rod manufacture
160 IF R$ = 'WILD' THEN F = 301.55 @ GOTO 190 !Wild rod offset = 301.55
170 IF R$ = 'KERN' THEN F = 301.50 @ GOTO 190 !Kern rod offset = 301.50
180 BEEP @ DISP 'NO SUCH ROD ID' @ GOTO 150
190 INPUT 'ENTER NETWORK ';N$ !Up to 15 characters
200 INPUT 'PERCENT CLOUD COVER ';N1$ !Enter as a whole number
210 INPUT 'WIND DIRECTION AND SPEED';W$ !Speed in MPH
220 INPUT 'HEIGHT OF GUN PERSON IN M ';H$
230 INPUT 'ENTER BOTH ROD #S ';R1$ !Use the serial number
240 DISP 'STADIA IMBALANCE IS';D2 @WAIT 1 !Imbalance if at a TBM else 0
250 DISP 'IS'; D2; ©INPUT 'CORRECT?,Y,N ';SS$ !Allows for wrong input
260 IF SS$ = 'Y' THEN Q = 0 @GOTO 280 ELSE 270
270 INPUT 'ENTER CORRECT IMBALANCE';Q
280 INPUT 'BS BENCH MARK';R2$ !Backsight bench mark identification
290 INPUT 'BS ROD ID ';R3$ !Serial number of rod on the BM
300 PRINT #1;DATE$ !Write the date to the file
310 PRINT #1;TIME$ !Write the time to the file
320 PRINT #1;G1$,G9$,T$,R$,N$,N1$
330 PRINT #1;W$,H$,R1$,R2$,R3$ !Write to the file
340 M = 0
350 X = X + 1 !Counter
360 INPUT 'BACKSIGHT LOW ';A !Backsight middle wire low scale
370 A1 = A + F !Backsight plus offset
380 I = 0 !Counter for stadia input
390 GOSUB 430 !Subroutine for backsight stadia
400 INPUT 'FORESIGHT LOW ';B !Foresight middle wire low scale
410 B1 = B + F !Foresight plus offset
420 GOTO 540 !Subroutine for foresight stadia
430 I = 0 !counter for times stadia is misread
440 I=I+1
450 IF M=1 THEN 400 ELSE 460 !Re-do reading, but not the stadia
460 IF I>2 THEN BEEP @ GOTO 360 ELSE 470 !You get two chances for stadia
470 INPUT 'BS TOP STADIA ';A2
480 INPUT 'BS BOTTOM STADIA ';A3
490 A4=A2-A @A5=A-A3 @A6=ABS(A4-A5) !Calculating backsight stadia
500 IF A6>P THEN BEEP @ GOTO 440 ELSE 510 !If greater then 0.5 m increment
510 A7=A4+A5 @ IF A7<0 THEN BEEP @ GOTO 440 ELSE 520 !Error trap
520 DISP 'BS STADIA IS';A7 @ WAIT 2 IDistance to the backsight rod
530 RETURN !If M=1 that means the stadias were
!read and were within tolerance;
!therefore read foresight high
540 I=0
550 I=1+1
560 IF M=1 THEN 700 ELSE 570 !Same as 450
570 IF I>2 THEN BEEP @ GOTO 400 ELSE 580 !Gives you two chances
580 INPUT ' FS TOP STADIA ';B2
590 INPUT ' FS BOTTOM STADIA ';B3
600 B4=B2-B @ B5=B-B3 @ B6=ABS(B4-B5) !Calculating the foresight
610 IF B6>P THEN BEEP @ GOTO 550 ELSE 620 !Checking for reading error
620 B7=B4+B5 @ IF B7<0 THEN BEEP @ GOTO 550 ELSE 630 !should never have
!negative stadia distance
630 G5=ABS(A7-B7) !Checking for set-up balance
640 IF G5>P THEN BEEP @ DISP G5 @ WAIT 2 @ GOTO 360 ELSE 650 !Calculating stadia
650 D=D+A7 @ D1=D1+B7 @D2=D1-D+Q
660 IF ABS(D2) > 10 THEN GOTO 670 ELSE 690 !Checking for stadia error
670 BEEP @ DISP ;D2;'STADIA' @ WAIT 2 @ D=D-A7 @ D1=D1-B7
680 D2=D2-(D1+D) @ GOTO 350 !If error in stadia reset the counter
690 DISP 'FS STADIA IS';B7 @ WAIT 2 !Display foresight stadia distance
700 INPUT ' FORESIGHT HIGH ';C
710 INPUT ' BACKSIGHT HIGH ';C1
720 C2=ABS(A1-B1) @ C3=ABS(C1-C)
730 C4=ABS(C2-C3)
740 IF C4>G THEN GOTO 970 ELSE C5=(B1+C)/2 !Checking the elevation
!difference between low and high scale
750 INPUT 'ENTER TEMP TOP, BOTTOM ';P$
760 D2$=STR$(D2)
770 D3=D+D1
780 DISP 'STADIA DIST IS ';D3 @ WAIT 2!Handy when looking for BM's
790 IF D2>0 THEN GOSUB 1120 ELSE GOSUB 1150
800 DISP MEM;'BYTES LEFT' !Amount of memory left.
810 WAIT 1
820 D4=(A1+C1)/2
830 A$=STR$(A) @ A7$=STR$(A7) !Converting real numbers to string
840 B$=STR$(B) @B7$=STR$(B7)
850 C$=STR$(C) @C1$=STR$(C1)
860  D3$=STR$(D3)
870  PRINT #1;A1$,C1$,A7$,B$,C$,B7$,P$  !Writing it to the file
880  E(X)=D4-C5 + E(X-1)  !Keeping track of the elevations
890  E$(X)$=STR$(E(X))$
900  DISP 'ELEV DIFF IS ';E$(X) @ PRINT #1;E$(X)$
910  WAIT 2
920  INPUT 'WAS FS ON A BM? Y, N ';E3$  !Option to input a bench mark
930  IF E3$='Y' THEN 990 ELSE 340  !If not a BM ask for bs low
940  ASSIGN #1 TO *  !Close the file
945  CFLAG -3  !Enables the time out function
950  DISP 'PROGRAM ENDED'
960  END
970  BEEP @DISP 'WARNING BAD READING ' @WAIT 2  !This where the program
980  @M=1 @GOTO 360  !Resets our counter
990  INPUT 'FS BENCH MARK ID ';E4$  !Allows for Bench mark input
1000  PRINT #1;E4$,D3$,D2$  !This is the total elev diff.
1010  WAIT 2  !Zeroing out the variables
1020  DISP 'STADIA DIST IS ';D3$  !Input the stadia distance
1030  WAIT 2
1040  DISP MEM;'BYTES LEFT'  !Indicating the amount of memory left
1050  WAIT 1
1060  DISP 'STADIA IMBALANCE IS ';D2  !Imbalance at Bench mark
1070  WAIT 1
1080  INPUT 'CONTINUE OR END? C, E ';E$  !If at a temporary Bench mark don't have to input all of the headings
1090  IF E$='E' THEN 940 ELSE INPUT 'WAS FS A BM OR A TBM?Y,N';S$  !If a Bench mark then go to the top
1100  IF S$='TBM' THEN 320 ELSE E(X)$=0 @D=0 @D1=0 @D2=0 @GOTO 120
1110  GOTO 1120
1120  DISP 'FORESIGHT IS';D2;'LONG'  !Indicates the stadia imbalance
1130  WAIT 2
1140  RETURN
1150  DIS\'FORESIGHT IS';ABS(D2);'SHORT'  !Indicates the stadia imbalance
1160  WAIT 2
1170  RETURN
1180  RETURN

**NOTE:** We have on several occassions accidently unpluged the HP-IL interface module while removing the cables after interfacing with the peripherals. This greatly disrupts the HP-71B's ability to communicate with the printer or cassette drive and must be restored. If this happens the following commands must be issued: Reset HPIL <End Line>, Restore IO <End Line>.
PROGRAM RECALL

10 !PROGRAM RECALL
20 ASSIGN #1 TO LEV !Opens file "LEV" in ram
30 X=0 !Counter for number of records printed
40 OFF ERROR !Error trap
50 ON ERROR GOTO 110
60 READ #1;N$ !Reads the first variable
70 DISPLAY IS :PRINTER !Prints the Data
80 DISP N$ !Also displays the data to the screen
90 X=X+1 !Increment the counter
100 GO TO 40 !Go back and get another data point
110 IF ERRN=31 THEN READ #1;N@DISP ERRM$ @ GOTO 120 ELSE 130
120 DISP N ©GOTO 40 IDisplay the culprit
130 IF ERRN=54 THEN 140 ELSE 40 !54=end of file
140 DISP ERRM$
150 WAIT 2
160 DISP X !Print number of records written
170 DISP TIMES $ !Print the time of day
180 END

Note: The reason for step number 110 is because we have had problems with unwanted characters being put into the file and causing and error when the file is read. This statement allows us to read around this problem.

PROGRAM TAPEWRT

10 !PROGRAM TAPEWRT
20 ASSIGN #1 TO LEV !Open file Lev in the HP71b's internal RAM
30 ASSIGN #2 TO LEV:TAPE !Opens file lev1 on the digital cassette
40 ASSIGN #3 TO COUNT !This is where the "Record number" is stored
50 INPUT 'FIRST WRITE TO TAPE? Y,N';A$
60 IF A$='Y' THEN 110
70 INPUT DO YOU REALLY MEAN YES? ';A1$
80 IF A1$='Y' THEN 50
90 J=0 !Resets the pointer to record "0"
100 GOTO 170 !Since this is the first write there is no "J"
110 READ #3;J !Read the counter number
120 DISP 'THE LAST RECORD NUMBER WAS ';J
130 WAIT 3
140 INPUT 'IS THIS THE CORRECT NUMBER TO WRITE? Y,N';A2$
150 IF A2$='Y' THEN 160 ELSE INPUT 'ENTER THE CORRECT RECORD # ';J
160 IF J>2700 THEN 340 ELSE 170 !If less than 2700 then start reading
else end the program
170 RESTORE #2,J !Restore the pointer to the correct position
180 ON ERROR GOTO 230
190 READ #1;N$ !Read the first data point from internal RAM
200 PRINT #2;J;N$ !Write the data to the cassette
210 J=J+1 !Increment the counter
220 IF J>2995 THEN 340 ELSE 180 !3000 is the maximum number of records
230 IF ERRN=31 THEN READ #1;N @ OFF ERROR @GOTO 180 ELSE 250
240 IF ERRN=54 THEN DISP ERRMS @GOTO 250!If end of file, end
250 J=J-1
260 DISP J
270 PURGE COUNT !Clear out the counter file
280 ASSIGN #3 TO COUNT !Open up the counter file
290 PRINT #3;J !Store the new counter number to memory
300 ASSIGN #1 TO * !Close all the files
310 ASSIGN #2 TO *
320 ASSIGN #3 TO *
330 GOTO 360
340 DISP 'RECORD LENGTH IS';J ©WAIT 2 !Display warning if nearing 3000
350 DISP 'USE ANOTHER TAPE' @WAIT 2
360 END

PROGRAM TAPEDUMP

10 ! PROGRAM TAPEDUMP
20 RESTORE IO !Restore the input/output status on the HP-IL
30 REMOTE !Allows the HP-IL device to go to remote mode
40 OUTPUT :RS232 ;"R0;SBC;P4;EE0;SW0;LE0;SL7" !Sets up baud rate, !parity, protocol etc. See HP-IL/RS232 !Interface manual
50 LOCAL !Sets the RS232C to local mode
60 DESTROY ALL !Destroy all variables
70 ASSIGN #2 TO LEV1:TAPE !Opens file LEV1 on the digital cassette tape
80 INPUT 'ENTER FIRST RECORD TO WRITE';! !Enter the first record number
90 RESTORE #2;J !Restore the pointer to the correct position
100 INPUT 'ENTER LAST RECORD TO WRITE';K !Last good record on tape
110 ON ERROR GO TO 170 !Error trap
120 READ #2;J;N$ !Read the first data point from the tape
130 J=J+1
140 OUTPUT :INTRFCE ;N$ !Send the data to the RS232
150 IF J=K THEN 190 !If less than 0 then end the program
160 GO TO 110 !Go back and get another data point
170 IF ERRN=31 THEN READ #2;J;N @ OFF ERROR @J=J+1 @ GOTO 110 ELSE 180
180 IF ERRN=54 THEN DISP J @ OFF ERROR @ J=J+1 @ GOTO 110 ELSE 190
190 DISP ERRMS$ !Display the error message
200 ASSIGN #2 TO * !Close the file and end the prog
210 END
Note: Statement number 170 is necessary to correct the problem with unwanted characters in the file. Statement number 180 is there because of possible blank spaces in the file that could be interpreted as an "End of file" markers.

PROGRAM TAPERD

10! PROGRAM TAPERD
20  ASSIGN #2 TO LEV1:TAPE  !Open the file on tape
30  INPUT 'FIRST RECORD TO READ';J  !Indicate where to start reading
40  RESTORE #2,J  !Restore the pointer
50  INPUT 'LAST RECORD TO READ';K  !Indicate the last record to read
60  OFF ERROR
70  ON ERROR GOTO 110
80  READ #2,J;N$  !Read the first record on tape
90  DISP N$ @J=J+1  !Display the value and increment the counter
100 IF J=K THEN GOTO 120 ELSE 60  !If J and K are equal then stop
110 IF ERRN=31 THEN READ #2,J;N @J=J+1 @GOTO 60 ELSE 120
120 IF ERRN=54 THEN DISP J @ J=J+1 @GOTO 60 ELSE 130
130 DISP ERRM$
140  ASSIGN #2 TO *
150  END
LIST OF VARIABLES FOR STADLEVE PROGRAM

A = Backsight middle wire low scale reading
A\textsubscript{1} = Backsight low scale + rod offset Wild=301.55, Kern=301.50
A\textsubscript{2} = Backsight top stadia wire
A\textsubscript{3} = Backsight bottom stadia wire
A\textsubscript{4} = Backsight top stadia interval
A\textsubscript{5} = Backsight bottom stadia interval
A\textsubscript{6} = Checking that difference between top and bottom stadia is < 0.5
A\textsubscript{7} = Stadia distance between instrument to backsight rod
B = Foresight middle wire low scale reading
B\textsubscript{1} = Foresight low scale reading + rod offset
B\textsubscript{2} = Foresight top stadia wire
B\textsubscript{3} = Foresight bottom stadia wire
B\textsubscript{4} = Foresight top stadia interval
B\textsubscript{5} = Foresight bottom stadia interval
B\textsubscript{6} = Difference between top, middle and bottom stadia wire < 0.5
B\textsubscript{7} = Stadia distance between instrument to foresight rod
C = Foresight middle wire high scale reading
C\textsubscript{1} = Backsight middle wire high scale reading
C\textsubscript{2} = Backsight low scale minus foresight low scale
C\textsubscript{3} = Backsight high scale minus foresight high scale
C\textsubscript{4} = Elevation difference between the low scale and high scale readings
C\textsubscript{5} = Average of the foresight low scale plus the foresight high scale
D = Cumulative backsight stadia reading
D\textsubscript{1} = Cumulative foresight stadia readings
D\textsubscript{2} = Stadia imbalance between the backsight and foresight distance
D\textsubscript{3} = Total stadia distance
D\textsubscript{4} = Average of the backsight low scale plus the backsight high scale
E = Cumulative elevation difference
E\textsubscript{3}\textdollar = Test for foresight bench mark
E\textsubscript{4}\textdollar = Foresight bench mark ID
E\textsubscript{5}\textdollar = Test for continuing or ending program
F = Either 301.550 or 301.5 depending on rod manufacturer
F\textdollar = Test to see if we really want to run the program
G = Limit for elevation difference between the low and high scale
G\textsubscript{5} = Checks that stadia balance does not exceed 5 meters
G\textsubscript{1}\textdollar = Serial number of the level gun
G\textsubscript{9}\textdollar = Instrument manufacture and model type
H\textdollar = Height of instrument person
I = Counter for stadia input
M = Counter for error between the low scale and high scale elevation
N\textdollar = Network
N\textsubscript{1}\textdollar = Percentage of cloud cover
P = Allowable error between the three stadia wires
P\textdollar = Temperature top and bottom
Q = Correction factor for stadia imbalance
R$ = Rod manufacture
R1$ = Rod serial numbers for both rods
R2$ = Backsight benchmark
R3$ = Backsight rod ID
S$ = Test for stadia inbalance input
S5$ = Variable to check for a TBM or BM
T$ = Units that the temperatures are recorded in
W$ = Wind speed and direction
X = Counter for elevation array
CONCLUSION:

Our experience has shown that surveying time in the field can be reduced by 30%, or more, when the notes are taken electronically as opposed to the traditional way of recording in a notebook. The use of a computer system can also be an appreciable time saver when transferring the data from the digital cassette recorder to an office computer, effectively eliminating the time-consuming process of inputting the data by hand. Computational errors in the field are virtually eliminated during the data-gathering process, thereby insuring the integrity of the field data, and input errors are also eliminated when direct transfer from the cassette recorder to the office computer is through the RS232 interface.

On the average, 2 weeks of field data can be reduced in about two or three days of office time. The man-hours saved by this system would certainly justify the $1500.00 cost, which could easily be recouped in only a few months.

ACKNOWLEDGEMENTS:

The author would like to thank all members of his survey crew for putting up with the long delays, innumerable "bugs", and many frustrating program changes (corrections?) that were necessary to get the programs to their present stage. Certainly the programs would not have been as "user friendly" if it were not for the many useful suggestions that these members made.

Special thanks to W. P. Kaiser of the U. S. Geological Survey's National Mapping Division, Vancouver, Washington, for his critique of the programs and his many valuable comments, to W. P. Johnson of the David A. Johnston Cascades Volcano Observatory (CVO) for his help in getting this paper into its published form, to P. T. Delaney of the Hawaiian Volcano Observatory for his critical review of this manuscript, and to W. C. Stokes of CVO for also being a reviewer.

REFERENCES:

NOAA, 1974, Classification, Standards of Accuracy, and General specifications of Geodetic Control Surveys. NOAA publication S/T 81-29, 12 pages.